
Miguel A. Gutiérrez Naranjo

Francisco J. Mart́ın Mateos

Dpto. de Ciencias de la Computación e Inteligencia Artificial

Universidad de Sevilla

IA2 99–00 CcIa Aplicaciones de SBC: Decisión y metaintérpretes 12.1



Decisión y metaintérpretes

x Árboles de decisión

x Razonamiento regresivo

x Bibliograf́ıa:

u Giarratano, J.C. y Riley, G. “Expert Systems Princi-

ples and Programming (2nd ed.)” (PWS Pub. Co.,

1994)

* Cap. 12.3: “Decision Trees”

* Cap. 12.4: “Backward Chaining”

u Lucas, P. y Gaag L.v.d. “Principles of expert sys-

tems” (Addison Wesley, 1991)

* Cap. 3: “Production rules and inference”

u Harmon, P. y King D. “Sistemas expertos” (Dı́az de

Santos, 1988)

Árboles de decisión

x Árboles de decisión binarios

RATON ARDILLA

¿ Emite chillidos ?

¿ Tiene cuello largo ?

JIRAFA ¿ Tiene trompa ?

¿ Le gusta estar en el agua ?ELEFANTE

HIPOPOTAMO RINOCERONTE

¿ Es grande ?

CHESKO

SI NO

NOSI

NOSI

NOSI

SI

NO

Árboles de decisión

x Sesión: Clasificación

CLIPS> (load "decision.clp")
%%************
TRUE
CLIPS> (reset)
CLIPS> (run)
- Es muy grande.? (si o no): si
- Tiene un cuello largo.? (si o no): no
- Tiene una trompa.? (si o no): si
Creo que es un/a elefante
¿Estoy en lo cierto.? (si o no): si

u Intentemos clasificar un gato

¿Lo intentamos de nuevo.? (si o no): si
- Es muy grande.? (si o no): no
- Emite chillidos.? (si o no): no
Creo que es un/a ardilla
¿Estoy en lo cierto.? (si o no): no

Árboles de decisión

x Sesión: Aprendizaje

¿Que animal es.?
gato
¿Que pregunta se debe contestar afirmativamente para

distinguir gato de ardilla.?
Maulla.?
Ahora ya puedo distinguir un/una gato

u El gato ya esta en la base de conocimiento

¿Lo intentamos de nuevo.? (si o no): si
- Es muy grande.? (si o no): no
- Emite chillidos.? (si o no): no
- Maulla.? (si o no): si
Creo que es un/a gato
¿Estoy en lo cierto.? (si o no): si
¿Lo intentamos de nuevo.? (si o no): no
CLIPS>

Árboles de decisión

x Hojas del árbol de decisión: Respuestas finales.

(deftemplate nodo-respuesta
(slot nombre)
(slot valor))

x Nodos internos del árbol de decisión: Pregun-
tas que conducen a la clasificación

(deftemplate nodo-decision
(slot nombre)
(slot pregunta)
(slot respuesta-si)
(slot respuesta-no))

x Cargar datos de un archivo

;;; REGLA: inicializacion
;;; SI
;;; no hay nodo raı́z
;;; ENTONCES
;;; cargamos los datos del fichero "animales.dat"
;;; e indica que el nodo actual es el nodo raı́z.

(defrule inicializacion
(not (nodo-decision (nombre raiz)))
=>
(load-facts "animales.dat")
(assert (nodo-actual raiz)))

Árboles de decisión

x Representación del ejemplo (animales.dat)

(nodo-decision (nombre raiz)
(pregunta "Es muy grande.?")
(respuesta-si nodo1) (respuesta-no nodo2))

(nodo-decision (nombre nodo1)
(pregunta "Tiene un cuello largo.?")
(respuesta-si nodo3) (respuesta-no nodo4))

(nodo-decision (nombre nodo2)
(pregunta "Emite chillidos.?")
(respuesta-si nodo5) (respuesta-no nodo6))

(nodo-respuesta (nombre nodo3) (valor jirafa))

(nodo-decision (nombre nodo4)
(pregunta "Tiene una trompa.?")
(respuesta-si nodo7) (respuesta-no nodo8))

(nodo-respuesta (nombre nodo5) (valor raton))

(nodo-respuesta (nombre nodo6) (valor ardilla))

(nodo-respuesta (nombre nodo7) (valor elefante))

(nodo-decision (nombre nodo8)
(pregunta "Le gusta estar en el agua.?")
(respuesta-si nodo9) (respuesta-no nodo10))

(nodo-respuesta (nombre nodo9) (valor hipopotamo))

(nodo-respuesta (nombre nodo10) (valor rinoceronte))

Árboles de decisión

x Tratamiento de los nodos internos
;;; REGLA: nodo-decision
;;; SI
;;; el nodo actual es un nodo decisión y
;;; no tenemos ninguna respuesta del usuario
;;; ENTONCES
;;; requerimos del usuario una respuesta a la
;;; pregunta que tiene asociada el nodo decisión
;;; actual y
;;; almacenamos la respuesta.

(defrule pregunta-en-nodo-decision
(nodo-actual ?nombre)
(nodo-decision (nombre ?nombre)

(pregunta ?pregunta))
(not (respuesta ?))
=>
(printout t "- " ?pregunta " (si o no): ")
(assert (respuesta (read))))

Árboles de decisión

x Respuesta incorrecta

;;; REGLA: respuesta-incorrecta
;;; SI
;;; la respuesta proporcionada por el usuario no es
;;; "si" ni "no"
;;; ENTONCES
;;; eliminamos la respuesta incorrecta y
;;; le indicamos al usuario que se ha equivocado.

(defrule respuesta-incorrecta
?respuesta <- (respuesta ~si&~no)
=>
(retract ?respuesta)
(printout t "Respuesta incorrecta." crlf))

Árboles de decisión

x Respuesta afirmativa en los nodos internos

;;; REGLA: procesa-si-en-nodo-decision
;;; SI
;;; el nodo actual es un nodo decisión y
;;; tenemos una respuesta afirmativa del usuario
;;; ENTONCES
;;; cambiamos el nodo actual de manera adecuada.

(defrule procesa-si-en-nodo-decision
?nodo <- (nodo-actual ?nombre)
(nodo-decision (nombre ?nombre)

(respuesta-si ?si))
?respuesta <- (respuesta si)
=>
(retract ?nodo ?respuesta)
(assert (nodo-actual ?si)))

Árboles de decisión

x Respuesta negativa en los nodos internos

;;; REGLA: procesa-no-en-nodo-decision
;;; SI
;;; el nodo actual es un nodo decisión y
;;; tenemos una respuesta negativa del usuario
;;; ENTONCES
;;; cambiamos el nodo actual de manera adecuada.

(defrule procesa-no-en-nodo-decision
?nodo <- (nodo-actual ?nombre)
(nodo-decision (nombre ?nombre)

(respuesta-no ?no))
?respuesta <- (respuesta no)
=>
(retract ?nodo ?respuesta)
(assert (nodo-actual ?no)))

Árboles de decisión

x Tratamiento de los nodos hoja

;;; REGLA: nodo-respuesta
;;; SI
;;; el nodo actual es un nodo respuesta y
;;; no tenemos ninguna respuesta del usuario
;;; ENTONCES
;;; el sistema indica al usuario la posible
;;; respuesta y
;;; se pregunta al usuario si la respuesta es
;;; correcta (según su criterio) y
;;; almacenamos la respuesta.

(defrule pregunta-en-nodo-respuesta
(nodo-actual ?nombre)
(nodo-respuesta (nombre ?nombre) (valor ?valor))
(not (respuesta ?))
=>
(printout t "Creo que es un/a " ?valor crlf)
(printout t "¿Estoy en lo cierto.? (si o no): ")
(assert (respuesta (read))))

Árboles de decisión

x Clasificación satisfactoria
;;; REGLA: procesa-si-en-nodo-respuesta
;;; SI
;;; el nodo actual es un nodo respuesta y
;;; tenemos una respuesta afirmativa del usuario
;;; ENTONCES
;;; pasamos a la fase en la que se pregunta al
;;; usuario si quiere continuar.

(defrule procesa-si-en-nodo-respuesta
?nodo <- (nodo-actual ?nombre)
(nodo-respuesta (nombre ?nombre))
?respuesta <- (respuesta si)
=>
(retract ?nodo ?respuesta)
(assert (pregunta-de-nuevo)))

Árboles de decisión

x Clasificación incorrecta
;;; REGLA: procesa-no-en-nodo-respuesta
;;; SI
;;; el nodo actual es un nodo respuesta y
;;; tenemos una respuesta negativa del usuario
;;; ENTONCES
;;; el sistema pasa a la fase de aprendizaje.

(defrule procesa-no-en-nodo-respuesta
?nodo <- (nodo-actual ?nombre)
(nodo-respuesta (nombre ?nombre))
?respuesta <- (respuesta no)
=>
(retract ?nodo ?respuesta)
(assert (reemplaza-nodo-respuesta ?nombre)))

Árboles de decisión

x ¿Quieres continuar?

;;; REGLA: pregunta-de-nuevo
;;; SI
;;; estamos en la fase en la que se pregunta al
;;; usuario si quiere continuar y
;;; no tenemos ninguna respuesta del usuario
;;; ENTONCES
;;; se pregunta al usuario si quiere realizar otra
;;; consulta y
;;; almacenamos la respuesta.

(defrule pregunta-de-nuevo
(pregunta-de-nuevo)
(not (respuesta ?))
=>
(printout t "¿Lo intentamos de nuevo.? (si o no): ")
(assert (respuesta (read))))

Árboles de decisión

x Se vuelve a hacer una consulta
;;; REGLA: una-vez-mas
;;; SI
;;; estamos en la fase en la que se pregunta al
;;; usuario si quiere continuar y
;;; tenemos una respuesta afirmativa del usuario
;;; ENTONCES
;;; se comienza de nuevo desde el nodo raı́z.

(defrule una-vez-mas
?fase <- (pregunta-de-nuevo)
?respuesta <- (respuesta si)
=>
(retract ?fase ?respuesta)
(assert (nodo-actual raiz)))

Árboles de decisión

x Se salvan los datos
;;; REGLA: no-mas-por-favor
;;; SI
;;; estamos en la fase en la que se pregunta al
;;; usuario si quiere continuar y
;;; tenemos una respuesta negativa del usuario
;;; ENTONCES
;;; se salvan los hechos que componen el árbol de
;;; decisión en el fichero "animales.dat"

(defrule no-mas-por-favor
?fase <- (pregunta-de-nuevo)
?respuesta <- (respuesta no)
=>
(retract ?fase ?respuesta)
(save-facts "animales.dat"))

Árboles de decisión

x Aprendizaje y modificación de los datos

u Árbol inicial:

CHESKORESPUESTA
INCORRECTA

u Árbol modificado:

CHESKO

PREGUNTA
PROPUESTA

RESPUESTA
ANTIGUA

RESPUESTA
NUEVA

SINO

Árboles de decisión

;;; REGLA: aprende
;;; SI
;;; estamos en la fase de aprendizaje
;;; ENTONCES
;;; el sistema pregunta al usuario la respuesta
;;; correcta y la pregunta que tiene asociada y
;;; modifica de manera conveniente el árbol de
;;; decisión.

(defrule aprende
?h <- (reemplaza-nodo-respuesta ?nombre)
?dato <- (nodo-respuesta (nombre ?nombre)

(valor ?valor))
=>
(retract ?h ?dato)
(printout t "¿Que animal es.?" crlf)
(bind ?animal (read))
(printout t

"¿Que pregunta se debe contestar afirmativamente"
" para distinguir " crlf)

(printout t " " ?animal " de " ?valor ".?" crlf)
(bind ?pregunta (readline))
(printout t

"Ahora ya puedo distinguir un/una " ?animal crlf)
(bind ?nodo1 (gensym*))
(bind ?nodo2 (gensym*))
(assert

(nodo-respuesta (nombre ?nodo1) (valor ?animal))
(nodo-respuesta (nombre ?nodo2) (valor ?valor))
(nodo-decision

(nombre ?nombre) (pregunta ?pregunta)
(respuesta-si ?nodo1) (respuesta-no ?nodo2))

(pregunta-de-nuevo)))

Razonamiento regresivo

x Forma de las reglas

(regla (si <variable> es|mayor|menor <valor>
{ y <variable> es|mayor|menor <valor> }*)

(entonces <variable> es <valor>))

x Objetivo inicial

(objetivo-inicial <variable>)

x Ejemplo (teatro.dat)

(regla (si distancia mayor 5000)
(entonces tipo-de-medio es coche))

(regla (si distancia mayor 1000 y
premura menor 15)

(entonces tipo-de-medio es coche))
(regla (si distancia menor 1000 y

premura mayor 15)
(entonces tipo-de-medio es andando))

(regla (si tipo-de-medio es coche y
teatro-centrico es si)

(entonces medio es taxi))
(regla (si tipo-de-medio es coche y

teatro-centrico es no)
(entonces medio es coche-propio))

(regla (si tipo-de-medio es andando y
tiempo es malo)

(entonces medio es andando-con-abrigo))
(regla (si tipo-de-medio es andando y

tiempo es bueno)
(entonces medio es andando))

(objetivo-inicial medio)

IA2 99–00 CcIa Aplicaciones de SBC: Decisión y metaintérpretes 12.20



R
a
zo

n
a
m

ie
n
to

re
g
re

si
v
o

x
E
je

m
p
lo

(t
e
a
tr

o
.d

a
t)

(
p
r
e
g
u
n
t
a

(
a
t
r
i
b
u
t
o

d
i
s
t
a
n
c
i
a
)

(
t
e
x
t
o

"
D
i
m
e

l
a

d
i
s
t
a
n
c
i
a

a
l

t
e
a
t
r
o

(
e
n

m
e
t
r
o
s
)
"
)
)

(
p
r
e
g
u
n
t
a

(
a
t
r
i
b
u
t
o

p
r
e
m
u
r
a
)

(
t
e
x
t
o

"
D
i
m
e

c
u
a
n
t
o
s

m
i
n
u
t
o
s

q
u
e
d
a
n

p
a
r
a

e
l

c
o
m
i
e
n
z
o
"
)
)

(
p
r
e
g
u
n
t
a

(
a
t
r
i
b
u
t
o

t
i
e
m
p
o
)

(
t
e
x
t
o

"
D
i
m
e

s
i

e
l

t
i
e
m
p
o

e
s

b
u
e
n
o

o
m
a
l
o
"
)
)

(
p
r
e
g
u
n
t
a

(
a
t
r
i
b
u
t
o

t
e
a
t
r
o
-
c
e
n
t
r
i
c
o
)

(
t
e
x
t
o

"
D
i
m
e

s
i

e
l

t
e
a
t
r
o

e
s
t
a

e
n

e
l

c
e
n
t
r
o

o
n
o
"
)
)

IA
2

99
–0

0
C

c
I a

A
p
li
ca

ci
on

es
d
e

S
B

C
:
D

ec
is

ió

Razonamiento regresivo

x Sesión:

CLIPS> (load "regresivo.clp")
+%%*****+**********
TRUE
CLIPS> (reset)
CLIPS> (run)
Dime la distancia al teatro (en metros)
700
Dime cuantos minutos quedan para el comienzo
40
Dime si el tiempo es bueno o malo
malo
Solución: medio : andando-con-abrigo
CLIPS> (facts)
f-0 (initial-fact)
...
f-12 (objetivo-inicial medio)
f-16 (atributo distancia 700)
f-19 (atributo premura 40)
f-20 (atributo tipo-de-medio andando)
f-24 (atributo tiempo malo)
f-25 (atributo medio andando-con-abrigo)
For a total of 11 facts.
CLIPS>

Razonamiento regresivo

x Módulo MAIN

(defmodule MAIN
(export deftemplate objetivo

atributo
regla))

(deftemplate MAIN::regla
(multislot si)
(multislot entonces))

(deftemplate MAIN::pregunta
(slot atributo)
(slot pregunta))

Razonamiento regresivo

;;; REGLA: inicial
;;; cargamos los hechos del fichero "reglas.dat".

(defrule MAIN::inicial
=>
(load-facts "teatro.dat"))

;;; REGLA: objetivo-inicial
;;; SI
;;; tenemos un atributo como objetivo inicial y
;;; no se ha obtenido su valor
;;; ENTONCES
;;; almacenamos el atributo como objetivo.

(defrule MAIN::objetivo-inicial
(objetivo-inicial ?nombre)
(not (atributo ?nombre ?))
=>
(assert (objetivo ?nombre)))

;;; REGLA: solucion
;;; SI
;;; tenemos un atributo como objetivo inicial y
;;; se ha obtenido su valor
;;; ENTONCES
;;; presentamos el atributo y el valor al usuario.

(defrule MAIN::solucion
(objetivo-inicial ?nombre)
(atributo ?nombre ?valor)
=>
(printout t "Solución: " ?nombre ": " ?valor crlf))

Razonamiento regresivo

;;; REGLA: intenta-regla
;;; SI
;;; tenemos cierto atributo como objetivo y
;;; dicho atributo no tiene asignado valor y
;;; existe una regla que le asigna cierto valor
;;; a dicho atributo en sus acciones
;;; ENTONCES
;;; incluimos como nuevo objetivo el primer
;;; atributo que aparece en las condiciones de
;;; la regla.

(defrule MAIN::intenta-regla
(objetivo ?nombre)
(not (atributo ?nombre ?))
(regla (si ?variable $?)

(entonces ?nombre es ?))
=>
(assert (objetivo ?variable)))

Razonamiento regresivo

;;; REGLA: pregunta-valor-atributo
;;; SI
;;; tenemos cierto atributo como objetivo y
;;; dicho atributo no tiene valor asociado y
;;; no hay reglas que concluyan algún valor
;;; para ese atributo y
;;; existe una pregunta asociada al atributo
;;; ENTONCES
;;; planteamos la pregunta al usuario y
;;; almacenamos su respuesta como valor asociado
;;; a dicho atributo.

(defrule MAIN::pregunta-valor-atributo
?objetivo <- (objetivo ?nombre)
(not (atributo ?nombre ?))
(not (regla (entonces ?nombre ? ?)))
?pregunta <- (pregunta (atributo ?nombre)

(texto ?texto))
=>
(retract ?objetivo ?pregunta)
(printout t ?texto crlf)
(assert (atributo ?nombre (read))))

Razonamiento regresivo

x Módulo RAZONA

(defmodule RAZONA
(import MAIN deftemplate objetivo

atributo
regla))

;;; REGLA: objetivo-satisfecho
;;; SI
;;; tenemos cierto atributo como objetivo
;;; y dicho atributo tiene asignado valor
;;; ENTONCES
;;; eliminamos el objetivo.

(defrule RAZONA::objetivo-satisfecho
(declare (auto-focus TRUE))
?objetivo <- (objetivo ?nombre)
(atributo ?nombre ?)
=>
(retract ?objetivo))

Razonamiento regresivo

;;; REGLA: regla-satisfecha
;;; SI
;;; la condición de una regla menciona un atributo y
;;; el atributo tiene asociado un valor y
;;; dicho valor cumple la condición de la regla
;;; ENTONCES
;;; eliminamos la regla e
;;; indicamos que el atributo que aparece en las
;;; acciones de la regla toma el valor indicado.

(defrule RAZONA::regla-satisfecha-es
(declare (auto-focus TRUE))
(atributo ?nombre ?v1)
?r <- (regla (si ?nombre es ?v1)

(entonces ?objetivo es ?v2))
=> (retract ?r)

(assert (atributo ?objetivo ?v2)))

(defrule RAZONA::regla-satisfecha-mayor
(declare (auto-focus TRUE))
(atributo ?nombre ?v1)
?r <- (regla (si ?nombre mayor ?v2&:(> ?v1 ?v2))

(entonces ?objetivo es ?v3))
=> (retract ?r)

(assert (atributo ?objetivo ?v3)))

(defrule RAZONA::regla-satisfecha-menor
(declare (auto-focus TRUE))
(atributo ?nombre ?v1)
?r <- (regla (si ?nombre menor ?v2&:(< ?v1 ?v2))

(entonces ?objetivo es ?v3))
=> (retract ?r)

(assert (atributo ?objetivo ?v3)))

Razonamiento regresivo

;;; REGLA: elimina-regla-no-util
;;; SI
;;; en la condición de una regla aparece cierto
;;; atributo y
;;; el atributo tiene asociado un valor y
;;; dicho valor no cumple la condición de la regla
;;; ENTONCES
;;; dicha regla nunca se va a satisfacer y la
;;; podemos eliminar.

(defrule RAZONA::elimina-regla-no-util-es
(declare (auto-focus TRUE))
(atributo ?nombre ?valor)
?r <- (regla (si ?nombre es ~?valor $?))
=>
(retract ?r))

(defrule RAZONA::elimina-regla-no-util-mayor
(declare (auto-focus TRUE))
(atributo ?nombre ?valor1)
?r <- (regla (si ?nombre mayor ?valor2

&:(not (> ?valor1 ?valor2)) $?))
=>
(retract ?r))

(defrule RAZONA::elimina-regla-no-util-menor
(declare (auto-focus TRUE))
(atributo ?nombre ?valor1)
?r <- (regla (si ?nombre menor ?valor2

&:(not (< ?valor1 ?valor2)) $?))
=>
(retract ?r))

Razonamiento regresivo

;;; REGLA: modifica-regla-util
;;; SI
;;; en la condición de una regla aparece cierto
;;; atributo en primer lugar y
;;; la regla tiene más condiciones y
;;; el atributo ya tiene asignado valor y
;;; dicho valor cumple la condición de la regla
;;; ENTONCES
;;; modificamos la regla eliminando la condición
;;; asociada al atributo pues se satisface.

(defrule RAZONA::modifica-regla-util-es
(declare (auto-focus TRUE))
(atributo ?nombre ?valor)
?r <- (regla (si ?nombre es ?valor y $?resto))
=>
(modify ?r (si $?resto)))

(defrule RAZONA::modifica-regla-util-mayor
(declare (auto-focus TRUE))
(atributo ?nombre ?valor1)
?r <- (regla (si ?nombre mayor ?valor2

&:(> ?valor1 ?valor2) y $?resto))
=>
(modify ?r (si $?resto)))

(defrule RAZONA::modifica-regla-util-menor
(declare (auto-focus TRUE))
(atributo ?nombre ?valor1)
?r <- (regla (si ?nombre menor ?valor2

&:(< ?valor1 ?valor2) y $?resto))
=>
(modify ?r (si $?resto)))

